

PESQUISA OPERACIONAL I

Prof. Dr. José Vicente Caixeta Filho


Depart. de Economia, Administração e Sociologia

ESALQ - Universidade de São Paulo

jose.caixeta@usp.br

**ARMAZENAGEM,
MANUSEIO E
ACONDICIONAMENTO DE
PRODUTOS**


EFEITO NO CUSTO LOGÍSTICO TOTAL A PARTIR DO MAIOR USO DE ESTOQUES, ARMAZÉNS E MANUSEIO DE PRODUTOS



ARMAZENAGEM DE PRODUTOS


Armazéns ou centrais de distribuição (CDs) executam um papel-chave para aumentar a eficiência da movimentação de mercadorias. Permitem a compensação eficaz dos custos de estocagem com menores custos de transporte, ao mesmo tempo que mantêm ou melhoram o nível de serviço.

PRINCIPAIS RAZÕES PARA UMA ORGANIZAÇÃO UTILIZAR ESPAÇO FÍSICO DE ARMAZENAGEM




- **reduzir custos de transporte e produção**
- **coordenar suprimento e demanda**
- **auxiliar o processo de produção**
- **auxiliar o processo de marketing**

PRINCIPAIS TIPOS DE DEPÓSITOS



- **espaço físico próprio**
- **aluguel de espaço de terceiros**
- **aluguel de facilidades**
- **estoque em trânsito**

FATORES MAIS COMUMENTE UTILIZADOS PARA AVALIAR A LOCALIZAÇÃO DE UM DEPÓSITO



- leis de zoneamento locais
- atitude da comunidade e do governo local com relação ao depósito
- custo para desenvolver e conformar o terreno
- custos de construção
- disponibilidade e acesso a serviços de transportes
- potencial para expansão
- disponibilidade, salários, ambiente e produtividade da mão-de-obra local
- taxas relativas ao local e à operação do armazém
- segurança do local (fogo, furto, inundação etc.)
- valor promocional do local
- taxas de seguro e disponibilidade de financiamento
- congestionamento de tráfego nas redondezas do local

PROBLEMAS DE LOCALIZAÇÃO

Fábricas

Depósitos

Cientes

1

Boa Esperança

2

Santo Lúcio

Luz do Norte

3

Aprazereres

Biribó Mirim

4

5

Lajeado

6

CUSTOS DE DISTRIBUIÇÃO (\$/t)

	Fáb. Sto. Lúcio	Fáb. Aprazeres	Dep. Boa Esperança	Dep. Luz do Norte	Dep. Biribó Mirim	Dep. Lajeado
Dep. Boa Esperança	0,5	---				
Dep. Luz do Norte	0,5	0,3				
Dep. Biribó Mirim	1,0	0,5				
Dep. Lajeado	0,2	0,2				
Cliente 1	1,0	2,0	---	1,0	---	---
Cliente 2	---	---	1,5	0,5	1,5	---
Cliente 3	1,5	---	0,5	0,5	2,0	0,2
Cliente 4	2,0	---	1,5	1,0	---	1,5
Cliente 5	---	---	---	0,5	0,5	0,5
Cliente 6	1,0	---	1,0	---	1,5	1,5

CAPACIDADE MENSAL DAS FÁBRICAS E DEPÓSITOS

Fábrica/Depósito	Capacidade (t)
Fáb. Sto. Lúcio	150.000
Fáb. Aprazeres	200.000
Dep. Boa Esperança	70.000
Dep. Luz do Norte	50.000
Dep. Biribó Mirim	100.000
Dep. Lajeado	40.000

NECESSIDADES MENSAS DOS CLIENTES

Cliente	Demanda (t)
1	50.000
2	10.000
3	40.000
4	35.000
5	60.000
6	20.000

$$\text{Min } \sum_{i=1}^2 \sum_{j=1}^4 d_{ij} x_{ij} + \sum_{i=1}^2 \sum_{k=1}^6 e_{ik} y_{ik} + \sum_{j=1}^4 \sum_{k=1}^6 f_{jk} z_{jk}$$

sujeito a

$$\sum_{j=1}^4 x_{ij} + \sum_{k=1}^6 y_{ik} \leq A_i, \text{ para } i=1,2$$

$$\sum_{i=1}^2 x_{ij} \leq B_j, \text{ para } j=1,\dots,4$$

$$\sum_{k=1}^6 z_{jk} - \sum_{i=1}^2 x_{ij} = 0, \text{ para } j=1,\dots,4$$

$$\sum_{i=1}^2 y_{ik} + \sum_{j=1}^4 z_{jk} = C_k, \text{ para } k=1,\dots,6$$

$$x_{ij}, y_{ik}, z_{jk} \geq 0, \text{ para } i=1,2; j=1,\dots,4; k=1,\dots,6$$

onde: d_{ij} = custo do transporte entre fábricas e depósitos;

e_{ik} = custo do transporte entre fábricas e clientes;

f_{jk} = custo do transporte entre depósitos e clientes;

x_{ij} = quantidades transportadas entre fábricas e depósitos;

y_{ik} = quantidades transportadas entre fábricas e clientes;


z_{jk} = quantidades transportadas entre depósitos e clientes;

A_i = capacidade das fábricas;

B_j = capacidade dos depósitos;

C_k = necessidades dos clientes.

UTILIZAÇÃO DE LINGUAGEM ESPECÍFICA PARA PROBLEMAS DE OTIMIZAÇÃO



IDE File Edit Search Windows Help

modelo-localizacao-armazens.gms

```
SETS
I  fabricas / STO-LUCIO, APRAZERES /
J  depositos / ESPERANCA, DO-NORTE, BIRIBO, LAJEADO /
K  clientes / C1*C6 / ;


PARAMETERS

A(I)  capacidade das fabricas
/     STO-LUCIO    150
      APRAZERES   200 /

B(J)  capacidade dos depositos
/     ESPERANCA   70
      DO-NORTE    50
      BIRIBO     100
      LAJEADO    40 /

C(K)  necessidades dos clientes
/     C1    50
      C2    10
      C3    40
      C4    35
      C5    60
      C6    20 / ;
```

TELA DE ENTRADA DO SOFTWARE/LINGUAGEM GAMS



IDE File Edit Search Windows Help

modelo-localizacao-armazens.gms modelo-localizacao-armazens.lst

```
---- VAR X  quantidades transportadas entre fabricas e depositos

                LOWER    LEVEL    UPPER    MARGINAL

STO-LUCIO.ESPERANCA  .          .          +INF     .
STO-LUCIO.DO-NORTE  .          .          +INF    0.200
STO-LUCIO.BIRIBO    .          .          +INF    0.500
STO-LUCIO.LAJEADO   .          40.000    +INF     .
APRAZERES.ESPERANCA .          .          +INF   999.500
APRAZERES.DO-NORTE  .          50.000    +INF     .
APRAZERES.BIRIBO    .          55.000    +INF     .
APRAZERES.LAJEADO   .          .          +INF     EPS

---- VAR Y  quantidades transportadas entre fabricas e clientes

                LOWER    LEVEL    UPPER    MARGINAL

STO-LUCIO.C1  .          50.000    +INF     .
STO-LUCIO.C2  .          .          +INF   999.000
STO-LUCIO.C3  .          .          +INF    0.800
STO-LUCIO.C4  .          .          +INF    0.500
STO-LUCIO.C5  .          .          +INF   999.000
STO-LUCIO.C6  .          20.000    +INF     .
```

TELA DE SAÍDA DO SOFTWARE/LINGUAGEM GAMS

SETS

I fabricas / STO-LUCIO, APRAZERES /
 J depositos / ESPERANCA, DO-NORTE, BIRIBO, LAJEADO /
 K clientes / C1*C6 /;

PARAMETERS

A(I) capacidade das fabricas
 / STO-LUCIO 150
 APRAZERES 200 /

B(J) capacidade dos depositos
 / ESPERANCA 70
 DO-NORTE 50
 BIRIBO 100
 LAJEADO 40 /

C(K) necessidades dos clientes
 / C1 50
 C2 10
 C3 40
 C4 35
 C5 60
 C6 20 /;

TABLE D(I,J) custo unitario de distribuicao entre fabricas e depositos

	ESPERANCA	DO-NORTE	BIRIBO	LAJEADO
STO-LUCIO	0.5	0.5	1.0	0.2
APRAZERES	1000	0.3	0.5	0.2;

TABLE E(I,K) custo unitario de distribuicao entre fabricas e clientes

	C1	C2	C3	C4	C5	C6
STO-LUCIO	1.0	1000	1.5	2.0	1000	1.0
APRAZERES	2.0	1000	1000	1000	1000	1000;

TABLE F(J,K) custo unitario de distribuicao entre depositos e clientes

	C1	C2	C3	C4	C5	C6
ESPERANCA	1000	1.5	0.5	1.5	1000	1.0
DO-NORTE	1.0	0.5	0.5	1.0	0.5	1000
BIRIBO	1000	1.5	2.0	1000	0.5	1.5
LAJEADO	1000	1000	0.2	1.5	0.5	1.5;

VARIABLES

X(I,J) quantidades transportadas entre fabricas e depositos
 Y(I,K) quantidades transportadas entre fabricas e clientes
 Z(J,K) quantidades transportadas entre depositos e clientes
 W custo total de transportes ;

POSITIVE VARIABLE X, Y, Z ;

EQUATIONS

CUSTO funcao objetivo
 PRODUCAO(I) producao das fabricas
 ENTRADEPO(J) quantidades que entram nos depositos
 SAIDEPO(J) quantidades que saem dos depositos
 DEMCLIE(K) quantidades demandadas pelos clientes ;

CUSTO .. W =E= SUM((I,J),D(I,J)*X(I,J))+
 SUM((I,K),E(I,K)*Y(I,K))+SUM((J,K),F(J,K)*Z(J,K));

PRODUCAO(I) .. SUM(J, X(I,J)) + SUM(K,Y(I,K)) =L= A(I) ;

ENTRADEPO(J) .. SUM(I, X(I,J)) =L= B(J) ;

SAIDEPO(J) .. SUM(K, Z(J,K)) - SUM(I, X(I,J)) =E= 0 ;

DEMCLIE(K) .. SUM(I, Y(I,K)) + SUM(J, Z(J,K)) =E= C(K);

MODEL LOCALIZA /ALL/ ;


SOLVE LOCALIZA USING LP MINIMIZING W ;

DISPLAY X.L, Y.L, Z.L;

Fábricas

Depósitos

Clientes



*Solução ótima para
o problema de transporte
(Custo mínimo = \$ 198.500)*

PREFERÊNCIAS DOS CLIENTES

Cliente	Fornecedor
1	Fáb. Sto. Lúcio
2	Dep. Boa Esperança
3	indiferente
4	indiferente
5	Dep. Luz do Norte
6	Dep. Lajeado <i>ou</i> Dep. Biribó Mirim

SETS

I fabricas / STO-LUCIO, APRAZERES /
 J depositos / ESPERANCA, DO-NORTE, BIRIBO, LAJEADO /
 K clientes / C1*C6 /;

PARAMETERS

A(I) capacidade das fabricas
 / STO-LUCIO 150
 APRAZERES 200 /

B(J) capacidade dos depositos
 / ESPERANCA 70
 DO-NORTE 50
 BIRIBO 100
 LAJEADO 40 /

C(K) necessidades dos clientes
 / C1 50
 C2 10
 C3 40
 C4 35
 C5 60
 C6 20 /;

TABLE D(I,J) custo unitario de distribuicao entre fabricas e depositos

	ESPERANCA	DO-NORTE	BIRIBO	LAJEADO
STO-LUCIO	0.5	0.5	1.0	0.2
APRAZERES	1000	0.3	0.5	0.2;

TABLE E(I,K) custo unitario de distribuicao entre fabricas e clientes

	C1	C2	C3	C4	C5	C6
STO-LUCIO	1.0	1000	1.5	2.0	1000	1.0
APRAZERES	2.0	1000	1000	1000	1000	1000;

TABLE F(J,K) custo unitario de distribuicao entre depositos e clientes

	C1	C2	C3	C4	C5	C6
ESPERANCA	1000	1.5	0.5	1.5	1000	1.0
DO-NORTE	1.0	0.5	0.5	1.0	0.5	1000
BIRIBO	1000	1.5	2.0	1000	0.5	1.5
LAJEADO	1000	1000	0.2	1.5	0.5	1.5;

TABLE PREFDEP(J,K) preferencia do cliente com relacao ao deposito

	C1	C2	C3	C4	C5	C6
ESPERANCA	1	0	1	1	1	1
DO-NORTE	1	1	1	1	0	1
BIRIBO	1	1	1	1	1	0
LAJEADO	1	1	1	1	1	0;

TABLE PREFFAB(I,K) preferencia do cliente com relacao a fabrica

	C1	C2	C3	C4	C5	C6
STO-LUCIO	0	1	1	1	1	1
APRAZERES	1	1	1	1	1	1;

VARIABLES

X(I,J) quantidades transportadas entre fabricas e depositos
 Y(I,K) quantidades transportadas entre fabricas e clientes
 Z(J,K) quantidades transportadas entre depositos e clientes
 W custo total de transporte
 PREF nivel total de custo e preferencia ;

POSITIVE VARIABLE X, Y, Z ;

EQUATIONS

PREFERENCIA custo e preferencia
 CUSTO custo total
 PRODUCAO(I) producao das fabricas
 ENTRADEPO(J) quantidades que entram nos depositos
 SAIDEPO(J) quantidades que saem dos depositos
 DEMCLIE(K) quantidades demandadas pelos clientes ;

PREFERENCIA .. $PREF = E = \text{SUM}((I,J), D(I,J) * X(I,J)) +$

$\text{SUM}((I,K), \text{PREFFAB}(I,K) * E(I,K) * Y(I,K)) +$
 $\text{SUM}((J,K), \text{PREFDEP}(J,K) * F(J,K) * Z(J,K)) ;$

CUSTO .. $W = E = \text{SUM}((I,J), D(I,J) * X(I,J)) + \text{SUM}((I,K), E(I,K) * Y(I,K)) +$
 $\text{SUM}((J,K), F(J,K) * Z(J,K)) ;$

PRODUCAO(I) .. $\text{SUM}(J, X(I,J)) + \text{SUM}(K, Y(I,K)) = L = A(I) ;$

ENTRADEPO(J) .. $\text{SUM}(I, X(I,J)) = L = B(J) ;$

SAIDEPO(J) .. $\text{SUM}(K, Z(J,K)) - \text{SUM}(I, X(I,J)) = E = 0 ;$

DEMCLIE(K) .. $\text{SUM}(I, Y(I,K)) + \text{SUM}(J, Z(J,K)) = E = C(K) ;$

MODEL LOCALIZA /ALL/ ;


SOLVE LOCALIZA USING LP MINIMIZING PREF ;

DISPLAY X.L, Y.L, Z.L, W.L, PREF.L;

Fábricas

Depósitos

Clientes



*Solução ótima, considerando as preferências dos clientes
(Custo mínimo = \$ 246.000)*

DECORRÊNCIAS DO FECHAMENTO DE DEPÓSITOS EXISTENTES

Depósito	Economia (\$)
Boa Esperança	10.000
Lajeado	5.000

CARACTERÍSTICAS DOS POSSÍVEIS NOVOS DEPÓSITOS

Depósito	Custo (\$)	Capacidade (t)
Jundilândia (<i>novo</i>)	12.000	30.000
Água Seca (<i>novo</i>)	4.000	25.000
Luz do Norte (<i>expansão</i>)	3.000	20.000

CUSTOS DE DISTRIBUIÇÃO COM A INCLUSÃO DOS POSSÍVEIS NOVOS DEPÓSITOS (\$/t)

	Fáb. Sto. Lúcio	Fáb. Aprazeres	Dep. Jundilândia	Dep. Água Seca
Dep. Jundilândia	0,6	0,4		
Dep. Água Seca	0,4	0,3		
Cliente 1	1,0	2,0	1,2	---
Cliente 2	---	---	0,6	0,4
Cliente 3	1,5	---	0,5	---
Cliente 4	2,0	---	---	0,5
Cliente 5	---	---	0,3	0,6
Cliente 6	1,0	---	0,8	0,9

SETS

I fabricas / STO-LUCIO, APRAZERES /
 J depositos / ESPERANCA, DO-NORTE, BIRIBO, LAJEADO, JUNDILANDIA, AGUA-SECA /
 K clientes / C1*C6 / ;

PARAMETERS

A(I) capacidade das fabricas
 / STO-LUCIO 150
 APRAZERES 200 /

B(J) capacidade dos depositos
 / ESPERANCA 70
 DO-NORTE 50
 BIRIBO 100
 LAJEADO 40
 JUNDILANDIA 30
 AGUA-SECA 25 /

C(K) necessidades dos clientes
 / C1 50
 C2 10
 C3 40
 C4 35
 C5 60
 C6 20 / ;

TABLE D(I,J) custo unitario de distribuicao entre fabricas e depositos

	ESPERANCA	DO-NORTE	BIRIBO	LAJEADO	JUNDILANDIA	AGUA-SECA
STO-LUCIO	0.5	0.5	1.0	0.2	0.6	0.4
APRAZERES	1000	0.3	0.5	0.2	0.4	0.3 ;

TABLE E(I,K) custo unitario de distribuicao entre fabricas e clientes

	C1	C2	C3	C4	C5	C6
STO-LUCIO	1.0	1000	1.5	2.0	1000	1.0
APRAZERES	2.0	1000	1000	1000	1000	1000 ;

TABLE F(J,K) custo unitario de distribuicao entre depositos e clientes

	C1	C2	C3	C4	C5	C6
ESPERANCA	1000	1.5	0.5	1.5	1000	1.0
DO-NORTE	1.0	0.5	0.5	1.0	0.5	1000
BIRIBO	1000	1.5	2.0	1000	0.5	1.5
LAJEADO	1000	1000	0.2	1.5	0.5	1.5
JUNDILANDIA	1.2	0.6	0.5	1000	0.3	0.8
AGUA-SECA	1000	0.4	1000	0.5	0.6	0.9 ;

VARIABLES

X(I,J) quantidades transportadas entre fabricas e depositos
 Y(I,K) quantidades transportadas entre fabricas e clientes
 Z(J,K) quantidades transportadas entre depositos e clientes
 W custo total de transportes
 BOA manutencao ou nao do deposito em boa esperanca
 LUZ expansao ou nao do deposito de luz do norte
 LAJEA manutencao ou nao do deposito de lajeado
 JUNDI construcao ou nao do deposito em jundilandia
 AGUA construcao ou nao do deposito em agua seca ;

POSITIVE VARIABLE X, Y, Z ;

BINARY VARIABLES BOA, LUZ, LAJEA, JUNDI, AGUA ;

EQUATIONS

CUSTO funcao objetivo
 PRODUCAO(I) producao das fabricas
 ENTRADEPO1(J) quantidades que entram no deposito de boa esperanca
 ENTRADEPO2(J) quantidades que entram no deposito de luz do norte
 ENTRADEPO3(J) quantidades que entram no deposito de biribo mirim
 ENTRADEPO4(J) quantidades que entram no deposito de lajeado
 ENTRADEPO5(J) quantidades que entram no deposito de jundilandia
 ENTRADEPO6(J) quantidades que entram no deposito de agua seca
 SAIDEPO(J) quantidades que saem dos depositos
 DEMCLIE(K) quantidades demandadas pelos clientes ;

CUSTO .. W =E= SUM((I,J),D(I,J)*X(I,J))+ SUM((I,K),E(I,K)*Y(I,K))+
 SUM((J,K),F(J,K)*Z(J,K)) - 10*(1 - BOA) - 5*(1 - LAJEA) + 3*LUZ +
 12*JUNDI + 4*AGUA ;

PRODUCAO(I) .. SUM(J, X(I,J)) + SUM(K, Y(I,K)) =L= A(I) ;
 ENTRADEPO1(J)\$ (ORD(J) EQ 1) .. SUM(I, X(I,J)) =L= BOA*B(J) ;
 ENTRADEPO2(J)\$ (ORD(J) EQ 2) .. SUM(I, X(I,J)) =L= B(J) + LUZ*20 ;
 ENTRADEPO3(J)\$ (ORD(J) EQ 3) .. SUM(I, X(I,J)) =L= B(J) ;
 ENTRADEPO4(J)\$ (ORD(J) EQ 4) .. SUM(I, X(I,J)) =L= LAJEA*B(J) ;
 ENTRADEPO5(J)\$ (ORD(J) EQ 5) .. SUM(I, X(I,J)) =L= JUNDI*B(J) ;
 ENTRADEPO6(J)\$ (ORD(J) EQ 6) .. SUM(I, X(I,J)) =L= AGUA*B(J) ;
 SAIDEPO(J) .. SUM(K, Z(J,K)) - SUM(I, X(I,J)) =E= 0 ;
 DEMCLIE(K) .. SUM(I, Y(I,K)) + SUM(J, Z(J,K)) =E= C(K) ;
 OPTION OPTCR = 0 ;
 MODEL LOCALIZA /ALL/ ;
 SOLVE LOCALIZA USING MIP MINIMIZING W ;
 DISPLAY X.L, Y.L, Z.L, BOA.L, LUZ.L, LAJEA.L, JUNDI.L, AGUA.L ;


Fábricas

Depósitos

Clientes

X

Boa Esperança



Solução para o problema de transporte, em função da nova estrutura de depósitos
(Custo mínimo = \$ 174.000)

LIÇÃO DE CASA:

Foram apresentadas três variações (casos) do modelo matemático de localização ótima, codificadas no GAMS (arquivos .gms). Pede-se: a) que os três casos sejam resolvidos com o GAMS e que se apresente a solução “por extenso” para cada um deles (quais depósitos deverão ser mantidos/fechados/instalados assim como os fluxos correspondentes e o valor do custo total das movimentações – anexe as saídas do GAMS que atestem tais resultados); b) escolha um dos três casos e monte o arquivo correspondente para o LINDO, assim como mostre a solução obtida (até 17/11, terça, 19h – não há limite para o número máximo de páginas)